Identifying FIRST. and LAST. Rows in By-Groups with PROC SQL
A popular DATA step technique that millions of SAS users need and find useful is the ability to identify the first and
last observation in by-groups. SAS performs by-group processing by creating two temporary variables for each BY
variable: a FIRST.variable and a LAST.variable. These temporary variables are populated with a value of 0 or 1
depending on the placement of each observation in the by-group. For example, when an observation is first in a
by-group, SAS automatically assigns a value of 1 to the FIRST.variable. For all other observations in the by-group,
SAS assigns a value of 0 to FIRST.variable. Similarly, if an observation is last in a by-group, SAS assigns a value of 1
to LAST.variable. For all other observations in the by-group a value of 0 is assigned to LAST.variable. It should be
noted that these two temporary variables are available for conditional processing in the DATA step, and are not
automatically added to the output data set.
It should be noted that although the ability to identify FIRST. and LAST. observations in by-groups has been
available to DATA step users for a long time, the same is not true for traditional SQL users. Essentially this meant
that whenever an SQL user needed the capability to identify FIRST. and LAST. rows in by-groups, they had little
choice but to use the DATA step programming technique. This typically was not an acceptable approach for the
SQL purist, so I knew I had to do something.
Over the next three years I spent much of my free time researching and experimenting with every conceivable
coding approach I could muster. Finally, I came up with an acceptable SQL solution that could emulate the features
found in the stalwart DATA step approach. The result is pure SQL code that uses a subquery construct. Now, SQL
users everywhere, and even users working with many of the leading RDBMS’, have a way to identify the FIRST. and
LAST. rows in by-groups.
This tip illustrates the FIRST. and LAST. coding technique using a single table MOVIES, as shown below. The Movies
table consists of 22 rows and six columns: title, category, studio, and rating are defined as character, and length
and year are defined as numeric. After the FIRST. and LAST. rows are identified, the rows are then concatenated.
MOVIES Table

Identifying FIRST. Row in By-groups
SQL Code
/********************************************************************************/
/** ROUTINE: FIRST-BY-GROUP-ROWS
**/
/** PURPOSE: Derive the first (min) row within each by-group using a subquery. **/
/********************************************************************************/
proc sql ;
create table first_bygroup_rows as
select rating,
title,
'FirstRow' as ByGroup
from movies M1
where title =
(select min(title)
from movies M2
where M1.rating = M2.rating)
order by rating, title ;

FIRST. Results

Identifying LAST. Row in By-groups
SQL Code
/*******************************************************************************/
/** ROUTINE: LAST-BY-GROUP-ROWS
**/
/** PURPOSE: Derive the last (max) row within each by-group using a subquery. **/
/*******************************************************************************/
create table last_bygroup_rows as
select rating,
title,
'LastRow' as ByGroup
from movies M1
where title =
(select max(title)
from movies M2
where M1.rating = M2.rating)
order by rating, title ;

LAST. Results

Concatenating FIRST. and LAST. Rows
SQL Code
/*******************************************************************************/
/** ROUTINE: CONCATENATE-FIRST-LAST-BY-GROUP-ROWS
**/
/** PURPOSE: Concatenate the results from the first (min) row and last (max) **/
/**
row within each by-group, and print.
**/
/*******************************************************************************/
create table first_last_rows as
select rating, title, bygroup
from first_bygroup_rows
UNION ALL
select rating, title, bygroup
from last_bygroup_rows ;
select * from first_last_rows ;
quit ;

FIRST. and LAST. Results Combined

For More Information:
If readers are interested in finding out how to identify the BETWEEN. rows, please send an email to
KirkLafler@cs.com and I’ll be happy to send the code. Readers can also find many more application programming
techniques in my SAS Press book, PROC SQL: Beyond the Basics Using SAS, Second Edition.

